数値積分～台形公式～

はしもとじょーじ　（岡山大学理学部地球科学科）

台形公式

t の関数 $f(t)$ の区間 $[a,b]$ における定積分 $I=\int_a^b f(t)dt$ を考える． $f(t)$ が複雑な関数でその原始関数を求めることが困難であるような場合、計算機を使って定積分の値を見積もることがおこなわれる．これを数値積分という．

ここで $f(t)$ は区間 $[a,b]$ において連続で、その値はいたるところで求めることができるものとする．定積分の単純な見積もり方のひとつは、区間 $[a,b]$ を多数の微小区間に分割し、それぞれの区間内において $f(t)$ を直線で近似して積分を見積もるものである． $f(t)$ を直線で近似した各区間はそれぞれ台形の形になるため、この方法は台形公式と呼ばれる．



区間 $[t_{i-1},t_i]$ の台形の面積は

$$\Delta S_i=\frac{f(t_{i-1})+f(t_i)}{2}(t_i-t_{i-1})$$

したがって、区間 $[a,b]$ を n 等分して各微小区間の台形の面積を足し算すると

$$\int_a^b f(t)dt\approx\sum_{i=1}^{n}\Delta S_i=\left(\frac{f(a)+f(b)}{2}+\sum_{i=1}^{n-1}f(t_i)\right)\cdot\Delta t$$

$$\Delta t = \frac{b-a}{n}、\quad t_0 = a、\quad t_n = b$$

となる．

台形公式を使って計算される値は、区間の分割数を増やしていくほどによりよい近似値を与えるようになる．しかしながら計算時間その他の制約もあるので、分割数をただ大きくすればよいというものでもない．要求される精度に見合う近似値を得るのに必要な分割数はどのように決定すればよいか、少ない計算でより精度の高い近似値を得る方法、などなど，参考文献１の４章にいろいろと書いてある．

プログラムの再利用性：サブルーチンの活用

数値積分の出番は多い．それぞれの問題についてプログラムをゼロから書くのは面倒であるだけでなく，間違える可能性も高い．簡単な作業で積分区間や被積分関数を変更して数値積分をおこなうことのできるプログラムがあれば，いろいろな積分区間／被積分関数についての数値積分を手軽におこなうことができる．

簡単にできるということは，便利であると同時に間違いを減らすことでもある．この間違いをなくすというのは非常に重要である．答えがわかっているものであれば、最後の答えを見てプログラムにバグがないかチェックすることができる．しかし実際には答えがわからないからこそ数値積分するのである．ちょっとしたタイプミスなどで間違いが生じていたとしても、そのことに気づかないということがしばしば起こりえる．こうした問題は、再利用性の高いプログラムを活用することである程度まで避けることができる．答えのわかっている問題に適用することで正しく動作することが確認されたプログラムを再利用すれば、そのプログラムについては正しく動作しているという(ある程度の)保証を得ることができる．

プログラムの再利用性を高めるには、サブルーチンを利用するのが常套手段である．以下は台形公式によって数値積分をおこなうプログラムをサブルーチンを使って作成した例である．積分区間は12行目と13行目の `xs` と `xe` で与えられ、被積分関数は `Subroutine Func(x,fx)` で計算される．このサブルーチン以外で被積分関数を計算することはないので、被積分関数を変更するには、この `Subroutine Func` だけを変更すればよい．

```
1         Program Integral
2         Implicit NONE
3
```

```
c -----
      Real*8  xs, xe
      Integer n
      Real*8  ss

      Real*8  ds, s0
c -----
c set 'sekibun-kukan'
      xs = 0.0D0
      xe = 1.0D0

c -----
 1    Continue

c input 'bunkatsu-su'
      Write(*,*) 'bunkatsu no kazu'
      Read(*,*) n
      If (n.le.0) then
          Stop
      End If

c calculation
      Call Daikei(xs,xe,n,ss)

c output
      Write(*,*) 'calculated', ss
      s0 = (acos(-1.0D0))/4.0D0
      ds = (1.0D0-(s0/ss))*100.0D0
      Write(*,*) 'analytic  ', s0
      Write(*,*) 'error %   ', ds

c goto start
      Go to 1

      End
```

```
      Subroutine Daikei(xs,xe,n,ss)
      Implicit NONE
c -----
      Real*8  xs, xe
      Integer n
c -----
      Real*8  ss
c -----
      Real*8  x, dx, fx
      Integer i
c -----
      dx = (xe-xs)/(dble(n))
      Call Func(xs,fx)
      ss = fx/2.0D0
      Call Func(xe,fx)
      ss = ss + (fx/2.0D0)
      If (n.eq.1) then
          Go to 200
      End If

      x  = xs
      Do 100 i = 1, n-1
        x  = x + dx
        Call Func(x,fx)
        ss = ss + fx
 100  Continue

 200  Continue
      ss = ss * dx

      Return
      End


      Subroutine Func(x,fx)
      Implicit NONE
c -----
```

```
      Real*8  x
c -----
      Real*8  fx
c -----
      fx  = 1.0D0/(x*x+1.0D0)

      Return
      End
```

課題１－２．数値積分

左辺の積分を数値的に計算して右辺と一致することを確認する.

$$\int_0^1 \frac{1}{x^2+1}dx = \frac{\pi}{4}$$

$$\int_0^1 \sqrt{x(1-x)}dx = \frac{\pi}{8}$$

$$\int_0^1 \frac{1}{x+\sqrt{1-x^2}}dx = \frac{\pi}{4}$$

$$\int_{-\infty}^{\infty} e^{-x^2}dx = \sqrt{\pi}$$

ここから先はちょっと工夫が必要.

$$\int_0^1 \frac{1}{\sqrt{x(1-x)}}dx = \pi$$

$$\int_0^1 \frac{\ln x}{1+x}dx = -\frac{\pi^2}{12}$$

参考文献

1．Press et al.（1992）Numerical Recipes in FORTRAN, 2nd ed., Cambridge Univ. Press.